
ОБОЗНАЧЕНИЯ 

x, y, z координаты, ось z направлена вертикально, [м]; 

, ,i j k
rr r

 орты координатных осей; 

h толщина пласта, [м] 

t время, [с]; 

µ динамическая вязкость жидкости, [Па·с]; 

υ  
вектор скорости фильтрации жидкости, с компонентами υx, 

υy, υz, [м/с]; 

K абсолютная проницаемость пористой среды, [м2]; 

f абсолютная пористость среды; 

ρ плотность, [кг/м3]; 

P давление, [Па]; 

σ=Kh/µ коэффициент гидропроводности; 

Q массовый расход, [кг/с]; 

q 
интенсивность отбора массы из единицы объема в единицу 

времени, [ρ/t]. 

ВВЕДЕНИЕ 

Реальные явления, происходящие в пластовой системе при извлече-
нии или закачке флюидов, весьма сложны. Это упругие и вязко-упругие 
деформации скелета пласта, колебательные явления различной частоты и 
интенсивности, изменения фазового и химического состава флюидов, из-
менения температурных и гидродинамических полей пластовой системы, 
изменение и перераспределение энергии, запасенной в пластовой системе. 

Во многих случаях для практических целей наибольший интерес 
представляет нестационарная гидродинамика пластовой системы. Это свя-
зано как с необходимостью предсказания поведения пластовой системы, 
так и использования данных, накопленных в процессе эксплуатации – де-
битов скважин, распределений давления, кривых восстановления давления 
и т. д. Поэтому во всех современных программных продуктах, предназна-
ченных для создания моделей месторождений, обязательно присутствует 
гидродинамический расчет. 

Гидродинамические процессы, происходящие в пластовой системе, 
скрыты от стороннего наблюдателя в недрах земли. Поэтому так важно ис-
пользование в учебном процессе современных программных продуктов, 
позволяющих рассчитывать и визуализировать нестационарную гидроди-
намику пластовой системы. Однако широкое использование этих программ 
в учебном процессе сдерживается рядом проблем, перечисленных далее в 
порядке убывания их значимости: 

1)  большая стоимость этих программ; 
2)  необходимость предварительного обучения преподавателей, что 

требует времени и финансов; 
3)  высокие требования, предъявляемые к компьютерам; 
4)  закрытый программный код; 
5)  отсутствие подробного описания численных методов, использо-

ванных при написании программ. 
При создании данного практикума принималось во внимание не толь-

ко вышесказанное, но и то, что практически на всех персональных компь-
ютерах, используемых в учебном процессе, установлен Excel, обладающий 
богатыми графическими средствами отображения числовых данных и 
имеющий Visual Basic for Application (VBA) в качестве языка программи-
рования. 

Программа, текст которой на языке VBA для Excel 2003 приводится 
ниже, может использоваться в следующих случаях: 

1) для демонстрации нестационарных гидродинамических процессов, 
происходящих при фильтрации несжимаемой жидкости в упругом порис-
том пласте. Прежде всего это изменение давления и положения линий тока 
при переводе части добывающих скважин в нагнетательные; 

2) при изучении численных методов (в частности, с использованием 
программы легко продемонстрировать явление численной неустойчиво-
сти); 

3) для получения практических навыков программирования на языке 
VBA в Excel 2003; 

4) в учебно-научном моделировании пластовой системы; 
5) в качестве самостоятельной работы. В процессе ее выполнения ре-

комендуется изучить влияние неоднородности пласта на перераспределе-
ние давления; смоделировать остановку скважины с получением кривой 
восстановления давления; используя реальные кривые восстановления 
давления, подобрать распределение гидропроводности; рассмотреть одно-
мерную постановку и реализовать неявный конечно-разностный метод. 

ГЛАВА 1. МАТЕМАТИЧЕСКАЯ ПОСТАНОВКА 

1.1. Основные допущения 

Рассматривается изотермическое движение однофазной жидкости, 
при этом влияние силы тяжести и капиллярных сил не учитывается.  

Основные допущения: 
• абсолютная проницаемость не зависит от времени; 
• вязкость жидкости постоянна; 
• толщина пласта везде одинакова; 
• для скорости фильтрации выполняется закон Дарси; 
• все параметры непрерывные и имеют непрерывные частные произ-

водные. 

1.2. Уравнения фильтрации жидкости в упругой пористой среде 

Выделим в пористой среде, содержащей жидкость, элементарный 
прямоугольный параллелепипед G (рис. 1) с ребрами ∆x, ∆y, ∆z и центром 
в точке (x, y, z). 

 z 

y 

x 

∆y 

∆z 

∆x 

  

Рис. 1 

Согласно закону сохранения вещества, изменение массы в этом объе-
ме связано с оттоком жидкости через границы объема и отбором (наличи-
ем внутренних стоков) жидкости непосредственно из объема некоторым 
способом: 

изменение массы + внутренний сток + поток через границы = 0. 
Отметим, что отток и отбор (в механике сплошной среды они обычно 

называются источниковыми членами) могут иметь как положительное, так 
и отрицательное значение. В последнем случае они будут являться прито-
ком через границы объема и, соответственно, источником, находящимся 
внутри объема. Рассмотрим каждую составляющую более подробно. 

Первая составляющая. Если в некоторый момент времени в элемен-
те G объемом ∆x∆y∆z масса жидкости равна: 

fzyx ρ∆∆∆ , 

то изменение этой массы за единицу времени будет: 

 
t

f
zyx

∂
ρ∂∆∆∆ . (1) 

Вторая составляющая. В единицу времени из рассматриваемого 
объема некоторым способом отбирается жидкость массой: 

 )( zyxq ∆∆∆ .  (2) 

Третья составляющая. Обозначим через S+ поверхность объема G, 
ориентированную с помощью выбора внешней нормали. Тогда поток век-

торного поля ρυ  через границы объема G будет равен: 

ρυ
S

dS
+

+
∫∫ . 

Для определения этой величины воспользуемся теоремой Остроград-
ского–Гаусса [1], согласно которой поток векторного поля через поверх-
ность элемента G равен интегралу по объему от дивергенции этого векто-
ра: 

 ( )∫∫∫∫∫ υρ=υρ
+

+

GS

dxdydzdivdS ,  (3) 

где zyx
div ZYX

∂
ρυ∂+

∂
ρυ∂+

∂
ρυ∂=υρ )()()(

)( . Применим к правой части 

уравнения (3) интегральную теорему о среднем [1]: 

( ) ( )ρυ ρυ( ) ,
G

div dxdydz div m x y z m G= ∆ ∆ ∆ ∈∫∫∫ , 

то есть интеграл по всему объему элемента G будет равен дивергенции в 
некоторой внутренней точке m, принадлежащей G, умноженной на объем 
элемента G. Таким образом, для элемента G та часть изменения массы в 
единицу времени, которая вызывается результирующим потоком жидкости 
через все его боковые грани, будет равна: 

 (ρυ)( )div x y z∆ ∆ ∆ .  (4) 

Окончательный баланс. Объединяя выражения (1), (2) и (4), полу-
чим, что уменьшение массы в указанном элементе за единицу времени со-
ставит: 

( )ρ
ρυ

f
x y z x y z div q

t

∂∆ ∆ ∆ = ∆ ∆ ∆ − −  ∂ . 

Устремляя размеры элемента G к нулю, получим уравнение неразрыв-
ности: 

 ( ) qdiv
t

f −υρ−=
∂
ρ∂

, (5) 

Известно [2], что в диапазоне давлений, характерном для пластовых 
условий, можно использовать для плотности линеаризованное уравнение: 

( )[ ]00 1 РРж −β+ρ=ρ , 

где ρ0 = const – значение плотности, соответствующее давлению P0 = const; 
βж = const – коэффициент упругости жидкости. Кроме того для пластового 
диапазона изменения давления изменение пористости хорошо описывается 
зависимостью: 

( )00 РРff c −β+= , 

где f0 = const – значение пористости, соответствующее давлению P0 = const; 
βс = const – коэффициент упругости скелета пласта. Так как βж и βс малы, то 
слагаемым, содержащим произведение βжβс можно пренебречь, и тогда 
можно записать: 

 ( )( )00000 РРfff cж −β+βρ+ρ≈ρ .  (6) 

Для пластового изменения давления можно предположить, что 
ρ ≈ ρ0 = const, то есть жидкость несжимаемая. Тогда с учетом (6) уравнение 
неразрывности (5) примет вид: 

 ( )
0

* ρ
−υ−=

∂
∂β q

div
t

P
, (7) 

где cжf β+β=β 0*  – упругоемкость пласта. Будем считать, что направления 

главных осей тензора абсолютной проницаемости совпадают с направле-
нием осей координат и выполняется закон Дарси, определяющий линей-
ную зависимость скорости фильтрации от градиента давления: 

 µ
P

Κυ = − ∇   (8) 

или 

z

PK

y

PK

x

PK z
Z

y
Y

x
X ∂

∂
µ

−=υ
∂
∂

µ
−=υ

∂
∂

µ
−=υ ;; , 

где k
z

P
j

y

P
i

x

P
P

rrr

∂
∂+

∂
∂+

∂
∂=∇  – градиент давления;  

=Κ

Z

Y

X

K

K

K

00

00

00

 – тензор абсолютной проницаемости пористой сре-

ды, причем Kx = Kx(x,y,z), Ky = Ky(x,y,z), Kz = Kz(x,y,z). 
Подставляя (8) в (7), получим уравнение, описывающее фильтрацию 

жидкости в упругой пористой среде: 

0
* ρ

−






 ∇
µ
Κ=

∂
∂β q

Pdiv
t

P
 

или 

0
* ρ

+
∂
∂β=









∂
∂

µ∂
∂+









∂
∂

µ∂
∂+









∂
∂

µ∂
∂ q

t

P

z

PK

zy

PK

yx

PK

x
zyx

. 

Упростим вид этого уравнения, считая, что тензор абсолютной прони-
цаемости является шаровым: 

















=Κ
K

K

K

00

00

00

, 

где K = K(x,y,z). Учитывая это, перепишем уравнение в следующем виде: 

 
0

* ρ
+

∂
∂β=









∂
∂

µ∂
∂+









∂
∂

µ∂
∂+









∂
∂

µ∂
∂ q

t

P

z

PK

zy

PK

yx

PK

x
. (9) 

1.3. Уравнения фильтрации для плоского течения 

Предположим, что абсолютная проницаемость не меняется по верти-
кали: K = K(x,y), а перепады давления по координате z равны нулю: 

0.
P

z

∂ =
∂  

Тогда уравнение фильтрации (9) упростится и будет описывать плос-
кое течение: 

 
0

* ρ
+

∂
∂β=









∂
∂

µ∂
∂+









∂
∂

µ∂
∂ q

t

P

y

PK

yx

PK

x
.  (10) 

Аналогичное по виду уравнение получается, если в исходном трех-
мерном уравнении (9) перейти к параметрам, осредненным по толщине 
пласта. Отметим, что для плоскорадиального течения уравнение (9) примет 
вид: 

0
*

1

ρ
+

∂
∂β=









∂
∂

µ∂
∂ q

t

P

r

PrK

rr
, 

где r – радиальная координата. 
Предположим, что источниковый член меняется по следующему закону: 





∈
∉

=
],[,

],[,0

21

21

ZZzq

ZZz
q

h
, 

причем Z2 – Z1 = h. Этому соответствует ситуация, когда забой вскрыл 
пласт толщиной h. Проинтегрируем уравнение (10) от Z1 до Z2: 

dz
q

t

P
dz

y

PK

yx

PK

x

Z

Z

h

Z

Z
∫∫ 









ρ
+

∂
∂β=

















∂
∂

µ∂
∂+









∂
∂

µ∂
∂ 2

1

2

1 0
* . 

С учетом того, что величины, независящие от z, выносятся за знак ин-
теграла, получим: 

dzq
t

P
h

y

PK

yx

PK

x
h

Z

Z

h∫ρ
+

∂
∂β=

















∂
∂

µ∂
∂+









∂
∂

µ∂
∂ 2

10
*

1
 

или: 

 dzq
t

P
hPgraddiv

Z

Z

h∫ρ
+

∂
∂β=σ

2

10
*

1
))(( ,  (11) 

где µ
=σ Kh

 – коэффициент гидропроводности. 

Работа скважины обычно характеризуется массовым расходом Q. По-
этому перейдем от qh, к Q. Для этого воспользуемся соотношением, связы-
вающим qh и Q: 

∫ ∫ =














S

Z

Z

h Qdsdzq
2

1

, 

где S – это поперечная площадь забоя. Применяя к левой части интеграль-
ную теорему о среднем [1], получим: 

SmdzmqSdsdzq
S

Z

Z

h

Z

Z

h ∈=













∫ ∫∫ ,)(

1

2

1

. 

Будем приближенно считать, что  

dzqdzmq
Z

Z

h

Z

Z

h ∫∫ ≈
2

1

2

1

)( , 

тогда 

S

Q
dzq

Z

Z

h =∫
2

1

. 

С учетом этого уравнение (11) примет вид: 

или 

0
*))((

ρ
+

∂β=σ

S

Q

t

P
hPgraddiv , 

где 

∂

∂
∂σ

∂
∂=σ

y

P

yx

P

x
Pgraddiv ))(( . 

1.4. Начальные и граничные условия 

Как известно, уравнения в частных производных могут иметь беско-
нечное количество решений. Чтобы выбрать среди этих решений единст-
венное, необходимо определить рассматриваемую область и в зависимости 
от типа уравнений задать граничные и начальные условия. Уравнение (12) 
имеет параболический тип, поэтому необходимы начальные условия в виде 
распределения параметров в рассматриваемой области в начальный момент 
времени и граничные условия на границе области. 

Расчетная область задается в виде прямоугольника: 

[ ]
[ ],,

,,

0

0

k

k

yyy

xxx

∈
∈

 

на границе области будут использоваться стационарные граничные усло-
вия первого рода: 

( ) ( ) ( ) ( ) [ ]kk yyyytyxPytyxP ,,,,,,, 02010 ∈α=α= , 

( ) ( ) ( ) ( ) [ ]kk xxxxtyxPxtyxP ,,,,,,, 0430 ∈α=α= . 

Начальные условия задаются в виде: 

0),,()0,,( 5 =α= tyxyxP . 

ГЛАВА 2. ЧИСЛЕННЫЙ МЕТОД 

2.1. Конечно-разностная аппроксимация 

При решении уравнения (12) конечно-разностным методом можно ис-
пользовать явную или неявную схему. Сведения об этих схемах можно 
найти в [3]. Уравнение (12) будем решать с использованием явной конеч-
но-разностной схемы, которая проще неявной в реализации, но в отличие 
от неявной схемы для устойчивости расчета должно выполняться ограни-
чение на шаг интегрирования по времени. 

Для перехода к конечным разностям покроем расчетную область рав-
номерной прямоугольной сеткой с числом узлов Nx*Ny: 

xkx NxxxNiixxx /)(,...,2,1,0, 00 −=∆=⋅∆+= , 

yky NyyyNkkyyy /)(,...,2,1,0, 00 −=∆=⋅∆+= , 

где ∆x и ∆y – шаги по x и y. Шаг по времени обозначим через ∆t, тогда: 

∑∆= jtt . 

Перейдем к сеточным функциям, используя следующие обозначения: 

( ) j
ikPjkiPtyxP == ,,),,( , 

( ) ikkiyx σ=σ=σ ,),( , 

( ) ( ) ( ) ( ) kk kyky 222111 , α=α=αα=α=α , 

( ) ( ) ( ) ( ) ii ixix 444333 , α=α=αα=α=α , 

ikkiyx 555 ),(),( α=α=α . 
Источниковые члены задаются в узлах сетки: 

( ) j
ikQjkiQtyxQ == ,,),,( . 

Переход к конечным разностям осуществляем по формулам: 

где 

2
,

2
1

2

1
1

2

1
±

±

±

±

σ+σ=σσ+σ=σ ikik

ik

kiik

ki
. 

Конечно-разностный аналог уравнения (12) примет вид: 

( ) +
∆

σ+σ−σ−σ
=

ρ
+

∆
−β

−−−++++

2

1
2
1

2
1

2
11

2
1

* x

PPPP

S

PP
h

ki

ki

j
ik

ki

j
ik

j
ki

kij
ik

ik

 

( )22
2

1

1

2

11

2

1

,, yxtO
y

PPPP j
ik

ik

j
ik

ik

j
ik

ik

j
ik

ik

∆∆∆+
∆

σ+σ−σ−σ
+

−−−+++

, 

где ( )22,, yxtO ∆∆∆  – ошибка аппроксимации за счет дискретизации, то есть 
точность решения уравнения (12) по времени будет иметь первую степень, а 
по пространству – вторую степень. Окончательная расчетная формула:  

 

1 1 1 1 1 1
1 2 2 2 2

2
*

1 1 1 1 1 1

2 2 2 2
2

0

σ σ σ σ

β

σ σ σ σ

ρ

j j j j
i k ik ik i k

i k i k i k i k
j j

ik ik

j j j j
ik ik ik ik jik ik ik ik

ik

P P P P
t

P P
h x

P P P P
Q

y S

+ −+ + − −
+

+ ++ + − −

 − − +
∆ = + + ∆



− − +
+ − ∆


 (13) 

Вид этой явной конечно-разностной схемы с разностями Вперед по 
Времени и Центральными по Пространству (ВВЦП) изображен на рис. 2. 

Pj
ik 

Pj+1
ik 

Pj
ik–1 

Pj
ik+1 

Pj
i–1k 

Pj
i+1k 

∆x 

∆y 

∆t 

∆x 

∆y 
 

Рис. 2. Схема ВВЦП 

Критерий устойчивости для схемы ВВЦП имеет вид [4]: 

 ( ) NykNxi
yx

hyx
t

ik

,...,0,,...,0,
)max(2 22

*
22

==
∆+∆σ

β∆∆δ≤∆ ,  (14) 

где коэффициент δ≤1 обеспечивает запас по устойчивости. 

Используя закон Дарси (8) по найденному полю давления в узлах сет-

ки, можно найти поле скоростей. Для этого во внутренних узлах сетки ис-

пользуется конечно-разностное представление уравнения (8) центральны-

ми разностями: 

y

PP

h
V

x

PP

h
V ikikik

iky
kikiik

ikx ∆
−σ−=

∆
−σ−= −+−+

2
,

2
1111

, 

а на границах используются направленные разности [5, с. 713]: 

x

PPP

h
V

x

PPP

h
V kNxkNxNxkNxk

Nxkx
kkkk

kx ∆
+−σ−=

∆
−+−σ−= −−

2

43
,

2

43 212100
0 , 

y

PPP

h
V

y

PPP

h
V iNyiNyiNyiNy

iNyy
iiii

iy ∆
+−σ

−=
∆

−+−σ−= −−

2

43
,

2

43 212100
0 . 

Известно, что в любой момент времени в каждой точке вектор скорости 
направлен по касательной к линии тока, поэтому с использованием поля скоро-
стей можно схематично показать положение линий тока. Для этого использует-
ся простейший алгоритм, согласно которому из каждого узла сетки выстраива-
ется отрезок, направление которого совпадает с вектором скорости, а его длина 

j
ikl  на каждом временном шаге j вычисляется по формуле: 

( ) ( )
jj

ik
ik

j
yik

j
xikj

ik t
l

VV
l ∆

+
=

)(max

22

. 

2.2. Учет источниковых членов 

При численной реализации уравнения (12) для упрощения алгоритма 
вместо источниковых членов, которые располагаются в определенных узлах 
сетки, можно задавать в этих узлах изменение давления по некоторому за-
кону. Этому изменению давления будет соответствовать некоторый дебит. 
Для нахождения этого дебита рассмотрим узел сетки с индексом ik (рис. 3), 
в котором есть источниковый член. Роль источника будет выполнять нагне-
тательная или добывающая скважина, находящаяся в точке ik. Естественно, 
что скважина имеет определенный размер и в точку поместиться не может, 

однако диаметр скважины значительно меньше размеров расчетной сетки и 
такое допущение оправдано. 

 

i–1 

2 1 

3 4 

k+1 

k 

k–1 
i+1 i 

∆x 

∆y 

∆x 

∆y 

  
Рис. 3 

Проинтегрируем уравнение (12) по площадке 1234 (рис. 3), проходя-
щей через середины сторон сетки: 

 ∫∫∫∫∫∫ ρ
+

∂
∂β=σ

1234 01234

*

1234

))(( dxdy
S

Q

P
hdxdyPgraddiv . (15) 

Для правой части уравнения (15) можно приближенно записать: 

*

1234 01234

* , (16) 

где 1+j
ikQ  – дебит скважины, находящейся в узле сетки с индексом ik в мо-

мент времени j+1. 
Напомним теорему Остроградского–Гаусса [1]: 

( )∫∫∫∫
++

+=⋅=
L

yx

LG

dlnjaniadlnadxdyadiv ),cos(),cos()(
rrrrrrr

, 

где G – площадь; L+ – ее замкнутая граница, с направленной наружу нор-
малью n

r ; dl  – элемент границы; ),cos( ni
rr

 и ),cos( nj
rr

 – направляющие коси-
нусы между ортами осей координат и нормалью. Применяя эту теорему к 
левой части уравнения (15), получим: 

=

где многоточие обозначает выражение, аналогичное предыдущему инте-
гралу. Из рис. 3 видно, что для каждого из отрезков 12, 23, 34, 41 будут 
выполняться соотношения: 

отрезок 12 – ),cos( ni
rr

 = cos(270°) = 0, ),cos( nj
rr

 = cos(180°) = –1, dl = dx; 

отрезок 23 – ),cos( ni
rr

 = cos(0°) = 1, ),cos( nj
rr

 = cos(90°) = 0, dl = dy; 

отрезок 34 – ),cos( ni
rr

 = cos(90°) = 0, ),cos( nj
rr

 = cos(0°) = 1, d l= –dx; 

отрезок 41 – ),cos( ni
rr

 = cos(180°) = –1, ),cos( nj
rr

= cos(90°) = 0, dl = –dy. 
Тогда предыдущее выражение примет вид: 

=

Переходя к численному интегрированию получим: 
2 3 3 4

1 2 4 1

Таким образом, объединяя (16) и (17), получим уравнение (15) в виде: 

откуда можно найти дебит: 

В программе, которая приведена далее, для каждой скважины задается 

изменение давления в зависимости от времени. Очевидно, что используя 

формулу (18), можно найти дебит этих скважин, что предлагается в каче-

стве самостоятельной работы для студентов. 

Отметим, что в [2] и [4] учет дебита скважины производится по дру-

гой формуле, в основу которой положена формула Дюпюи для радиально-

го течения в окрестности скважины. 

ГЛАВА 3. АЛГОРИТМ 

В процессе отладки алгоритма выяснилось, что расчетная часть VBA 
программы, связанная с вычислением и отображением на диаграммах дав-
лений и скоростей, выполняется значительно быстрее, чем создание и раз-
мещение диаграмм на листе Excel. Поэтому программа разделена на две не 
связанные между собой части – первая часть создает, масштабирует, раз-
мещает диаграммы и указывает расположение данных для них, а вторая 
часть рассчитывает искомые величины и выводит их на диаграммы (см. 
рис. 4). Каждая из этих частей запускается нажатием соответствующей 
кнопки, о чем пойдет речь ниже. 

 

 

Расчет параметров и 
вывод их на 
диаграммы 

 
Создание диаграмм 

 
Рис. 4 

Для упрощения алгоритма контроль за последовательностью действий 
возлагается на пользователя. Расчет параметров и вывод их на диаграммы 
без предварительного создания диаграмм может привести к ошибке в VBA. 

Последовательность действий при создании диаграмм отражена на 
рис. 5. При размещении 3D диаграммы для давления выполняются сле-
дующие действия: 
1) указывается, что диаграмма помещается на лист Excel, указывается 

имя этого листа; 
2) диаграмма сдвигается на 3000 единиц влево и на 200 единиц вверх; 
3) указывается диапазон данных, которые будут использоваться для по-

строения диаграммы; 
4) указывается тип диаграммы «поверхность»; 
5) удаляется «легенда»; 
6) указывается название диаграммы и шрифт; 
7) название диаграммы перемещается на 325 единиц вправо; 
8) указывается максимальное значение «Pнагнет»; 
9) указывается минимальное значение «Pдобыв»; 
10) удаляются оси «X», «Y» и горизонтальные линии. 

При размещении диаграммы для линий тока выполняются следующие 
действия: 
1) находятся минимальные и максимальные значения координат узлов 

сетки, общее количество графиков; 
2) указывается, что диаграмма помещается на лист Excel, указывается 

имя этого листа; 
3) диаграмма сдвигается на 3000 единиц влево и на 135 единиц вниз; 

4) указывается начало данных и что данные находятся в строках; 
5) указывается тип диаграммы «линейный без маркеров»; 
6) удаляется «легенда»; 
7) указывается название диаграммы и шрифт; 
8) название диаграммы перемещается на 325 единиц вправо; 
9) указывается минимальное и максимальное значение по оси «X»; 
10) указывается минимальное и максимальное значение по оси «Y»; 

 
Nx*Ny>210 

Диаграммы 
имеются ? 

 

Присвоение исходных данных, граничных и 
начальных условий, рассылка данных для 
диаграмм, размещение 3D диаграммы для 

давления и плоской диаграммы для линий тока 

Удаление 
диаграмм 

да 

да 

нет 

нет 

Стоп 

 
Рис. 5. Создание диаграмм 

11) удаляются оси «X», «Y», горизонтальные линии и границы области 
построения; 

12) закраска внутренней области устанавливается белой; 
13) для первого графика указываются данные «Xi», «Yi», которые будут 

использоваться для построения рамки, ограничивающей рассматри-
ваемую область; 

14) для первого графика назначается синий цвет; 
15) толщина графика – средняя; 
16) линия – сплошная; 
17) далее, NGR раз в пп. 18–20 стоятся линии, выходящие из узлов сетки, 

с длиной пропорциональной скорости и направлением, совпадающим 
с направлением скорости. Число линий (графиков) равно числу узлов 
сетки NGR; 

18) создается очередной линейный график; 
19) для него на листе lin  указывается положение данных – начало графика 

(x1,y1) и конец (x2,y2); 
20) указывается, что график рисуется сплошной, средней толщины, крас-

ной линией; 
21) после окончания построения диаграммы с нее снимается выделение. 

Алгоритм расчета параметров и вывод их на диаграммы состоит из 
следующих шагов: 
1) задаются исходные данные, начальные и граничные условия; 
2) далее, Nt раз выполняются пп. 3–10; 
3) по формуле (13) рассчитывается поле давления во вспомогательном 

массиве; 
4) пересылка из вспомогательного массива в основной массив; 
5) вычисляется время; 
6) задается давление на забое скважин; 
7) по найденному полю давления с использованием центральных разностей 

вычисляется поле скоростей. По полю скоростей строятся отрезки, вы-
ходящие из узлов сетки и совпадающие по направлению с линиями тока, 
а по длине пропорциональные скорости в этой точке; 

8) координаты начала и конца каждого отрезка заносятся на лист lin , 
найденное давление заносится на лист pres, номер расчетного шага и 
рассчитанное время заносятся на лист all; 

9) опрашивается кнопка, ответственная за остановку расчета: если кноп-
ка нажата, то выводится сообщение с предложением или продолжить 
расчет, или закончить его; 

10) если выбирается продолжение расчета, то выполняется пункт 3, если 
окончание расчета, то расчет прекращается. 

ГЛАВА 4. ПРОГРАММА 

Для решения уравнения (12) с использованием явной конечно-
разностной схемы (13) воспользуемся программой, написанной на Visual 
Basic for Application (VBA) для Microsoft Excel 2003. Файл oil01.xls с про-
граммой выложен в открытый доступ на сайте научно-технической биб-
лиотеки Томского политехнического университета по адресу: 
http://www.lib.tpu.ru/fulltext/m/2008/oil01.xls. 

Отметим, что при создании этой программы широко использовался 
режим записи макросов с последующим редактированием полученного 
VBA текста. Первичные сведения по VBA можно найти в [6], [7], [8] или в 
другой литературе, посвященной использованию VBA в Excel. 

4.1. Особенности использования программы в Excel 

Перед открытием файла необходимо в Excel установить возможность 
работы с макросами: сервис → макрос → безопасность → закладка 
«Уровень безопасности» → средняя. После этого, при открытии файла 
oil01.xls указать: не отключать макросы.  

При самостоятельной модификации программы возможно появление 
в ней ошибок, из-за которых программа зацикливает. Чтобы остановить ее, 
необходимо нажать одновременно две клавиши Ctrl  и Pause. Далее, в от-
крывшемся окне нажать кнопку End. Таким же способом можно в любой 
момент времени остановить программу. 

Если нет возможности скачать файл, то в этом случае необходимо 
выполнить пошаговые инструкции, которые приводятся далее. 

4.2. Последовательность действий при создании программы 

После запуска Excel установить, как указано выше, средний уровень 
безопасности. 

Убедиться, что в открытой книге Excel имеется три листа. Если мень-
ше, то добавить: поставить указатель мыши (далее просто указатель) на 
ярлык листа, нажать правую кнопку мыши и в контекстном меню выбрать 
добавить (далее правой кнопкой выбрать…). 

Переименовать названия листов в all, lin , pres: поставить указатель на 
ярлык листа, правой кнопкой выбрать переименовать, набрать all, нажать 
Enter. Так же переименовать другие два листа в lin  и pres. 

Перейти в редактор VBA и создать модуль для набора текста про-
граммы: нажать Alt + F11, строка меню →→→→ Insert →→→→ Module. 

Набрать текст программы, который приведен в разделе «Текст про-
граммы». 

Убедиться в отсутствии грубых ошибок: строка меню →→→→ Debug →→→→ 
Compile. Если сообщения нет, то грубые ошибки отсутствуют, если сооб-
щение есть, то необходимо исправить ошибку и повторить Debug →→→→ 
Compile. 

Заметим, что далее слова процедура и макрос являются синонимами. 
Перейти в Excel: Alt + F11. 
Перейти на лист all. 
Указатель мыши поставить в строку меню, правой кнопкой вызвать 

контекстное меню, выбрать Формы, в появившемся меню нажать изобра-
жение кнопки (рис. 6). 

 

  
Рис. 6 Рис. 7 

 

  
Рис. 8 Рис. 9 

При нажатой правой кнопке мыши нарисовать кнопку левее столбца O 
и ниже строки 4. После того, как правая кнопка мыши отпущена, возникнет 
меню Назначить макрос объекту (рис. 7), в котором перечислены все дос-
тупные для Excel процедуры VBA. Необходимо выбрать процедуру полеPU 
и нажать OK . Далее необходимо поместить указатель мыши внутри создан-
ной кнопки и изменить имеющийся текст на слово Расчет. 

Повторяя последние два раздела аналогичным образом создать кноп-
ку Остановка, назначить ей макрос остановка; создать кнопку Создает, 
назначить ей макрос создает. 

Снять выделение с созданной кнопки, щелкнув по любой ячейке. 
Перед началом работы нажать кнопку Создает для создания диа-

грамм на листе all, отображающих процесс расчета (рис. 8). Эту же кнопку 
необходимо нажимать перед началом расчета, если изменились какие-либо 
параметры в программе. 

Расчет запускается нажатием кнопки Расчет. При этом на листе lin  
будут создаваться данные для построения линий тока, а на листе pres – 
данные для построения поля давления. Результаты будут отображаться на 
диаграммах на листе all. 

Приостановка расчета происходит при нажатии кнопки Остановка. 
После чего необходимо выбрать или завершение расчета – кнопка ДА или 
продолжение расчета – кнопка НЕТ (рис. 9). 

4.3. Особенности программной реализации 

Укажем некоторые особенности используемых конструкций VBA. 
В программе используется инструкция Option Explicit , поэтому все 

переменные должны быть описаны. 
Для продолжения текста программы на следующую строку необходи-

мо набрать два символа: пробел и подчерк. 
Далее знак : разделяет операторы на одной строке 

а = b + 1 : d = c + d 
При описании переменных следующие две конструкции эквивалентны 

между собой: 
Dim a%, b#, c$, d 
Dim a As Integer, b As Double, c As String, d As Variant  

Эквивалентная предыдущим строкам форма записи: 
Dim a As Integer 
Dim b As Double 
Dim c As String 
Dim d 

Если тип не указан, считается, что тип As Variant. 

Необходимо помнить, что следующие две строки неэквивалентны: 
Dim x%, y%, z%  
Dim x,    y,     z% 

Так как некоторые функции есть в Excel, но отсутствуют в VBA, то 
используются следующие конструкции: 
Application.Min(x1,x2…)  – возвращает минимальное значение из x1,x2… 
Application.Max(x1,x2…) – возвращает максимальное значение из x1,x2… 
Application.Pi( ) – возвращает число π. 

Функция MsgBox используется в двух вариантах: 
MsgBox(“Text”)  – выводит Text и кнопку OK, 
i=MsgBox(“Text”, vbYesNo) – выводит Text и две кнопки ДА, НЕТ. i=6 
при нажатой кнопке ДА, i=7 при нажатой кнопке НЕТ. 

После апострофа идет комментарий, то есть текст, невлияющий на ра-
боту программы. Комментарии можно опустить, поэтому далее две строки 
эквивалентны:  
a(i, k) = 1 
a(i, k) = 1     ' это коэффициент 

В программе используются три процедуры (макросы), которые дос-
тупны в Excel: 
Public Sub создает – удаляет предыдущие диаграммы, создает начальные 
заготовки диаграмм для линий тока и давления, задает начальные и гра-
ничные условия в начальный момент времени. Эту процедуру необходимо 
выполнять, если в программе изменились какие-либо данные, а также при 
первом запуске программы. 
Public Sub полеРU – выполняются шаги по времени, при этом производится 
расчет давления и вывод результатов на лист pres, расчет скорости и вывод 
результатов на лист lin , вывод результатов на готовые диаграммы на листе 
all, изменение режима работы скважин, вывод рассчитанного времени. 
Public Sub остановка – приостанавливает работу и выводит сообщение с 
двумя кнопками ДА и НЕТ. При нажатии кнопки ДА работа программы 
продолжается, при нажатии кнопки НЕТ работа программы заканчивается. 

Отметим, что служебное слово Public может быть опущено. 
Кроме того в программе описаны 7 процедур со служебным словом 

Private, которые не доступны в Excel, а используются в указанных выше 
процедурах: 
Private Sub нач_дан – подготавливает данные для работы программы: ко-
эффициент гидропроводности (в программе постоянная величина), пло-
щадь сечения забоя, дебит скважин, длина шагов dx и dy. Производит очи-
стку листов lin , pres и записывает на них данные, находит шаг по времени 
из условия устойчивости (14). 
Private Sub нач_усл – задает начальное распределение давления. 

Private Sub гр_усл – задет граничные условия. 
Private Sub скважины – задает положение скважин, давление на забое и в 
зависимости от времени переводит добывающие скважины в нагнетательные. 
Private Sub полеU – расчет компонент вектора скорости, масштабирова-
ние и рассылка данных для диаграмм. 
Private Sub диагр_давл – создает 3D диаграмму для давления. 
Private Sub диагр_лин_тока – создает плоскую диаграмму для линий тока. 

Для уменьшения длины текста программы в двух последних процеду-
рах используются переменные As Variant. Поэтому выражения: 
Dim m2 
Set m2=ActiveSheet 
m2.Shapes(N).IncrementLeft-3000 
m2.Shapes(N).IncrementTop 135 
эквивалентны конструкциям: 
ActiveSheet.Shapes(N).IncrementLeft-3000 
ActiveSheet.Shapes(N).IncrementTop 135. 

Кроме того, используются операторы With … End With . Поэтому ин-
струкции: 
With m1.SeriesCollection(i+1) 
 .XValues=Range(m3.Cells(Lx,1),m3.Cells(Lx,2)) 
 .Values=Range(m3.Cells(Ly,1),m3.Cells(Ly,2)) 
 .Border.ColorIndex=3 
 .Border.Weight=xlMedium 
 .Border.LineStyle=xlContinuous. 
End With 
эквивалентны следующим инструкциям: 
m1.SeriesCollection(i+1).XValues=Range(m3.Cells(Lx,1),m3.Cells(Lx,2)) 
m1.SeriesCollection(i+1).Values=Range(m3.Cells(Ly,1),m3.Cells(Ly,2)) 
m1.SeriesCollection(i+1).Border.ColorIndex=3 
m1.SeriesCollection(i+1).Border.Weight=xlMedium 
m1.SeriesCollection(i+1).Border.LineStyle=xlContinuous. 

Глобальные переменные и константы, используемые в программе, на-
ходятся между строкой Option Explicit  и первой процедурой. Они имеют 
следующее назначение: 
Nt – число шагов по времени, которое ограничивает расчет; 
Lx, Ly  – размеры рассчитываемой области; 
Nx, Ny – количество узлов сетки по каждой координате; 
Pнач, Pнагнет, Pдобыв – пластовое давление, давление на забое нагнета-
тельной и добывающей скважины; 
h, betta, d, po – толщина пласта, упругоемкость пласта, диаметр забоя, 
плотность жидкости; 
dt, dx, dy, tt, S – шаги по времени, по координатам, рассчитанное время, 
площадь забоя. 

Глобальные массивы имеют следующее назначение: 
P1 – давление в узлах сетки на текущем временном слое; 
a – гидропроводность в узлах сетки; 
X – абсциссы узлов сетки; 
Y – ординаты узлов сетки; 
Q – дебиты скважин в узлах сетки. 

В процессе расчета на лист all в ячейки K1 и K2 выводятся соответ-
ственно число шагов по времени и расчетное время в часах. Поле давлений 
отображается на объемной диаграмме, а для демонстрации линий тока ис-
пользуется точечная диаграмма со значениями, соединенными отрезками 
без маркеров. Из-за ограниченного числа графиков на одной точечной диа-
грамме в программе имеется ограничение на число точек сетки: 
Nx*Ny<210. 

4.4. Текст программы 

Option Explicit 
 
Const Nt% = 100, Lx# = 2000, Ly# = Lx, Nx% = 14, Ny% = 14, Pнач# = 200e5 
Const Pнагнет# = 250e5, Pдобыв# = 100e5, h# = 10, betta# = 1e-9, d# = 0.15, po# = 900 
Dim dt#, dx#, dy#, tt#, S#, P1#(0 To Nx, 0 To Ny), a#(0 To Nx, 0 To Ny) 
Dim X#(0 To Nx, 0 To Ny), Y#(0 To Nx, 0 To Ny), Q#(0 To Nx, 0 To Ny) 
 
Public Sub создает( ) ' создает нач. заготовки диаграмм для  линий тока и давления 
Dim i%                       ' описание целой переменной 
 i = ActiveSheet.ChartObjects.Count                       ' i=количество диагр. на листе 
 If (i > 0) Then ActiveSheet.ChartObjects.Delete    'если i>0 то все диагр. удаляются 
 If (Nx * Ny > 210) Then        ' если число узлов сетки NX * NY > 210 то 
  MsgBox ("Nx*Ny>210 !!!") ' вывод сообщения 
  End                                         'останов расчета 
 End If 
 Call исх_дан          ' исходные данные 
 Call нач_усл          ' исходные условия P 
 Call гр_усл            ' граничные условия Р 
 Call полеU(0, 0)    ' расчет, рассылка данных для диаграмм 
 Call диагр_давл     ' создает 3D диаграмму для давления 
 Call диагр_лин_тока ' создает диаграмму для линий тока 
End Sub 'создает( ) 
 
Public Sub полеРU( )   'расчет P + U,вывод результатов на готовые графики 
Dim t%, i%, k%, sx1#, sx2#, sy1#, sy2#, P2#(1 To Nx – 1, 1 To Ny – 1) 
 Call исх_дан      ' исходные данные 
 Call нач_усл      ' начальные условия P 
 Call гр_усл        ' граничные условия 
 For t = 0 To Nt   ' шаги по времени 
  For i = 1 To Nx – 1 

   For k = 1 To Ny – 1 ' расчет поля давления во вспомогательном массиве P2 
    sx1 = (a(i + 1, k) + a(i, k)) / 2: sx2 = (a(i, k) + a(i – 1, k)) / 2 
    sy1 = (a(i, k + 1) + a(i, k)) / 2: sy2 = (a(i, k) + a(i, k – 1)) / 2 
    P2(i, k) = P1(i, k) + dt / h / betta * ( _ 
     (P1(i + 1, k) * sx1 – P1(i, k) * (sx2 + sx1) + P1(i – 1, k) * sx2) / dx ^ 2 + _ 
     (P1(i, k + 1) * sy1 – P1(i, k) * (sy2 + sy1) + P1(i, k – 1) * sy2) / dy ^ 2 + _ 
     Q(i, k) / S / po) 
   Next k 
  Next i 
  For i = 1 To Nx – 1   ' P2 -> P1 
   For k = 1 To Ny – 1 
    P1(i, k) = P2(i, k) 
   Next k 
  Next i 
  tt = tt + dt               ' рассчитанное время 
  Call скважины(t)  ' давление на скважинах 
  Call полеU(t, tt)    ' расчет, рассылка данных для рисунков 
  DoEvents               ' ждет окончания отрисовки и нажатия кнопок 
 Next t                      ' конец шагов по времени 
End Sub    ' полеРU( ) 
 
Private Sub исх_дан( ) ' нач. парам. + исходные данные 
Dim i%, k%, L%, b#                      ' описание переменных 
 S = Application.Pi( ) * (d / 2) ^ 2 ' площадь сеч. забоя 
 tt = 0                                              ' текущее время 
 dx = Lx / Nx: dy = Ly / Ny:                    ' длина шагов по X и Y 
 Sheets("lin").Range("a:aw") = Empty     ' очистка листа lin 
 Sheets("pres").Range("a:aw") = Empty  ' очистка листа pres 
 Sheets("all").Cells(1, 12) = "№ шага по времени" ' вывод текста 
 Sheets("all").Cells(2, 12) = "время в часах"     ' вывод текста 
 Sheets("all").Cells(1, 11) = 0                             ' № временного шага 
 Sheets("all").Cells(2, 11) = tt                             ' время в часах 
 L = 1 
 For k = 0 To Ny 
  For i = 0 To Nx 
   a(i, k) = 1e-8                   ' коэфф гидропров. 
   Q(i, k) = 0                       ' дебит скважин 
   X(i, k) = dx * i                ' X узлов  сетки 
   Y(i, k) = dy * k               ' Y узлов  сетки 
   Sheets("lin").Cells(L, 1) = X(i, k)       'сетка для графика "скорость" 
   Sheets("lin").Cells(L + 1, 1) = Y(i, k) 'сетка для графика "скорость" 
   L = L + 2 
  Next i 
 Next k 
 For i = 0 To Nx                   ' сетка для графика "давление" 
  Sheets("pres").Cells(1, i + 2) = X(i, 0) 

 Next i 
 For k = 0 To Ny 
  Sheets("pres").Cells(k + 2, 1) = Y(0, k) 
 Next k 
 b = a(0, 0)                                    'находим шаг по времени из усл. уст. 
 For i = 0 To Nx 
  For k = 0 To Ny 
   b = Application.max(a(i, k), b) ' максимальный коэф. гидропров. 
  Next k 
 Next i 
 dt = 0.9 * h * betta / (2 * b * (1 / dx ^ 2 + 1 / dy ^ 2)) 'шаг по времени 
End Sub 'исх_дан( ) 
 
Private Sub скважины(t%) 'перевод из добыв. в нагнет. зависит от времени 
Dim i%, k%, i1%, k1%, i2%, k2%, i3%, k3%, i4%, k4%, t0% 
t0 = Nt / 2                'время перевода из добывающих в нагнетательные 
i = Fix(Nx / 2): k = Fix(Ny / 2)                  'коорд. добыв №0 
i1 = Fix(Nx / 4):     k1 = Fix(Ny / 4)          'коорд. добыв – нагнет №1 
i2 = Fix(3 * Nx / 4): k2 = Fix(Ny / 4)        'коорд. добыв – нагнет №2 
i3 = Fix(3 * Nx / 4): k3 = Fix(3 * Ny / 4)  'коорд. добыв – нагнет №3 
i4 = Fix(Nx / 4):     k4 = Fix(3 * Ny / 4)    'коорд. добыв – нагнет №4 
If t < t0 Then                't <= t0 
 P1(i1, k1) = Pдобыв   ' добыв 
 P1(i2, k2) = Pдобыв   ' добыв 
 P1(i3, k3) = Pдобыв   ' добыв 
 P1(i4, k4) = Pдобыв   ' добыв 
 P1(i, k) = Pдобыв      ' добыв 
Else               't >= t0 
 P1(i1, k1) = Pнагнет   ' нагнет 
 P1(i2, k2) = Pнагнет   ' нагнет 
 P1(i3, k3) = Pнагнет   ' нагнет 
 P1(i4, k4) = Pнагнет   ' нагнет 
 P1(i, k) = Pдобыв       ' добыв 
End If 
End Sub 'скважины( ) 
 
Private Sub нач_усл( ) ' начальные условия P 
Dim i%, k% 
 For i = 0 To Nx 
  For k = 0 To Ny 
   P1(i, k) = Pнач 
  Next k 
 Next i 
End Sub     'нач_усл( ) 

Private Sub гр_усл( )  'граничные условия P 
Dim i%, k% 
 For i = 0 To Nx 
  P1(i, 0) = Pнач:  P1(i, Ny) = Pнач 
 Next i 
 For k = 0 To Ny 
  P1(0, k) = Pнач:  P1(Nx, k) = Pнач 
 Next k 
End Sub 'гр_усл( ) 
 
Private Sub полеU(t%, tt#) ' расчет Ux,Uy, масштабирование. рассылка Ux,Uy,P 
Dim i%, k%, L%, b# 
Dim Ux#(0 To Nx, 0 To Ny), Uy#(0 To Nx, 0 To Ny) ' UX, UY 
 For i = 1 To Nx – 1 'расчет числителя для Ux,Uy. центральные разности 
  For k = 1 To Ny – 1 
   Ux(i, k) = P1(i + 1, k) – P1(i – 1, k) 
   Uy(i, k) = P1(i, k + 1) – P1(i, k – 1) 
  Next k 
  Ux(i, 0) = P1(i + 1, 0) – P1(i – 1, 0)          ' на границе 
  Ux(i, Ny) = P1(i + 1, Ny) – P1(i – 1, Ny) ' на границе 
 Next i 
 For k = 1 To Ny – 1 '  на границах 
   Uy(0, k) = P1(0, k + 1) – P1(0, k – 1) 
   Uy(Nx, k) = P1(Nx, k + 1) – P1(Nx, k – 1) 
 Next k 
 For k = 0 To Ny 'на границах, направленные разн. 
  Ux(0, k) = -3# * P1(0, k) + 4# * P1(1, k) – P1(2, k) 
  Ux(Nx, k) = 3# * P1(Nx, k) – 4# * P1(Nx – 1, k) + P1(Nx – 2, k) 
 Next k 
 For i = 0 To Nx ' на границах, направленные разн. 
  Uy(i, 0) = -3# * P1(i, 0) + 4# * P1(i, 1) – P1(i, 2) 
  Uy(i, Ny) = 3# * P1(i, Ny) – 4# * P1(i, Ny – 1) + P1(i, Ny – 2) 
 Next i 
 b = 0 
 For i = 0 To Nx    '   расчет Ux*dt, Uy*dt, max 
  For k = 0 To Ny 
   Ux(i, k) = -Ux(i, k) * a(i, k) / (2 * dx) * dt 
   Uy(i, k) = -Uy(i, k) * a(i, k) / (2 * dy) * dt 
   b = Application.max(Sqr(Ux(i, k) ^ 2 + Uy(i, k) ^ 2), b) ' max 
  Next k 
 Next i 
 If b = 0 Then b = 1 
 L = 1          ' рассылка давл. и проекций вект. скор. для рисунков 
 For k = 0 To Ny 
  For i = 0 To Nx 
   Sheets("lin").Cells(L, 2) = X(i, k) + Ux(i, k) * dx / b 

   Sheets("lin").Cells(L + 1, 2) = Y(i, k) + Uy(i, k) * dy / b 
   Sheets("pres").Cells(k + 2, i + 2) = P1(i, k) 
   L = L + 2 
  Next i 
 Next k 
 Sheets("all").Cells(1, 11) = t                ' вывод № шага по времени 
 Sheets("all").Cells(2, 11) = tt / 60 / 60 ' вывод рассчитанного времени в часах 
End Sub 'полеU( ) 
 

Private Sub диагр_давл( )   ' создает 3D диаграмму для давления 
Dim i%, L%, N$, m1, m2, m3 
 Sheets("all").Activate    ' положение диаграммы 
 Set m2 = ActiveSheet      ' текущий лист 
 Set m3 = Sheets("pres")   ' лист с данными для давления 
 Charts.Add.Location Where:=xlLocationAsObject, Name:=m2.Name ' диагр. на лист 
 N = m2.ChartObjects(1).Name ' имя диаграммы 
 Set m1 = ActiveChart               ' активная диаграмма 
 m2.Shapes(N).IncrementLeft -3000   ' сдвиг диагр. влево 
 m2.Shapes(N).IncrementTop -200    ' сдвиг диагр. вверх 
 m1.SetSourceData Source:=Range(m3.Cells(1, 1), m3.Cells(Ny + 2, Nx + 2))  'данные 
 m1.ChartType = xlSurface              'график "поверхность" 
 m1.HasLegend = False                    'нет легенды 
 m1.HasTitle = True                         'есть названия диагр 
 m1.ChartTitle.Characters.Text = "Давление" 'название диагр 
 Selection.Font.Name = "Times New Roman"    'шрифт 
 Selection.Font.Size = 8                                       'размер шрифта 
 m1.ChartTitle.Left = 325                                    'перемещает название 
 m1.Axes(xlValue).MaximumScale = Pнагнет 'масштаб max по вертикали 
 m1.Axes(xlValue).MinimumScale = Pдобыв  'масштаб min по вертикали 
 m1.Axes(xlCategory).Delete                      'удаляет ось X 
 m1.Axes(xlSeries).Delete                           'удаляет ось Y 
 m1.Axes(xlValue).MajorGridlines.Delete  'удаляет горизонтальные линии 
End Sub 'диагр_давл( ) 
 

Private Sub диагр_лин_тока( )              ' создает диаграмму для линий тока 
Dim i%, k%, NGR%, Lx%, Ly%, N$, m1, m2, m3 
Dim Xmax#, Xmin#, Ymax#, Ymin#     ' min,max для диаграммы 
 Xmax = X(Nx, Ny): Xmin = X(0, 0) 
 Ymax = Y(Nx, Ny): Ymin = Y(0, 0) 
 NGR% = (Nx + 1) * (Ny + 1)  ' общее кол-во графиков 
 Sheets("all").Activate         ' положение диагр 
 Set m2 = ActiveSheet         ' текущий лист 
 Set m3 = Sheets("Lin")      ' лист с данными для линий тока 
Charts.Add.Location Where:=xlLocationAsObject, Name:=m2.Name 'диагр на лист 
 N = m2.ChartObjects(2).Name  ' имя диаграммы 
 Set m1 = ActiveChart                ' активная диаграмма 
 m2.Shapes(N).IncrementLeft -3000   ' сдвиг диагр. влево 

 m2.Shapes(N).IncrementTop 135     ' сдвиг диагр. вниз 
 m1.SetSourceData Source:=m3.Cells(1, 1), PlotBy:=xlRows ' данные X – в строках 
 m1.ChartType = xlXYScatterLinesNoMarkers   'график "линейный без маркеров" 
 m1.HasLegend = False                    ' нет легенды 
 m1.HasTitle = True                         ' есть названия диагр 
 m1.ChartTitle.Characters.Text = "Скорость"    ' название диагр 
 Selection.Font.Name = "Times New Roman"    ' шрифт 
 Selection.Font.Size = 8                                       ' размер шрифта 
 m1.ChartTitle.Left = 325                                    ' перемещает название 
 m1.Axes(xlCategory).MinimumScale = Xmin   ' масштаб min X 
 m1.Axes(xlCategory).MaximumScale = Xmax  ' масштаб max X 
 m1.Axes(xlValue).MinimumScale = Ymin         ' масштаб min Y 
 m1.Axes(xlValue).MaximumScale = Ymax        ' масштаб max Y 
 m1.Axes(xlCategory).Delete               ' удаляет ось X 
 m1.Axes(xlValue).Delete                    ' удаляет ось Y 
 m1.Axes(xlValue).MajorGridlines.Delete  'удаляет горизонтальные линии 
 m1.PlotArea.Border.LineStyle = xlNone    'удаляет границы внутр.области 
 m1.PlotArea.Interior.ColorIndex = 2          ' внутр.область белая 
 With m1.SeriesCollection(1)                       ' график номер 1 
  .XValues = Array(Xmin, Xmax, Xmax, Xmin, Xmin) ' Xi рамка 
  .Values = Array(Ymin, Ymin, Ymax, Ymax, Ymin)    ' Yi рамка 
  .Border.ColorIndex = 5                       ' 5 – синий цвет 
  .Border.Weight = xlMedium 'xlThin  ' толщина 
  .Border.LineStyle = xlContinuous      ' сплошной 
 End With 
 For i = 1 To NGR ' место, где будут данные для графиков 2,3,...NGR 
  Lx = 2 * i – 1        ' строка, где находятся x1,x2 
  Ly = Lx + 1          ' строка, где находятся y1,y2 
  m1.SeriesCollection.NewSeries      ' создание заготовок для графиков 
  With m1.SeriesCollection(i + 1)      ' график номер i+1 
   .XValues = Range(m3.Cells(Lx, 1), m3.Cells(Lx, 2)) ' x1,x2 
   .Values = Range(m3.Cells(Ly, 1), m3.Cells(Ly, 2))    ' y1,y2 
   .Border.ColorIndex = 3                   ' 3 – красный цвет графиков 
   .Border.Weight = xlMedium           ' толщина графика 
   .Border.LineStyle = xlContinuous  ' сплошной график 
  End With 
 Next i 
m2.Cells(1, 1).Select     ' снимает выделение с диаграммы 
End Sub                        ' диагр_лин_тока( ) 
 

Public Sub остановка( )   'вывод сообщения с двумя кнопками 
Dim i%                 'продолж.или оконч.работы 
 i = MsgBox("Что делать дальше? Продолжить – Да, закончить – Нет", vbYesNo) 
 If (i = 7) Then End           ' нажали кнопку "Нет" – окончание работы 
End Sub 

ГЛАВА 5. ИСХОДНЫЕ ДАННЫЕ 

В программе использовались следующие исходные данные, которые 
являются характерными для нефтяных месторождений: 

пластовое давление – Pнач = 200е5 Па, 
давление на забое нагнетательной скважины – Рнагнет = 250е5 Па, 
давление на забое добывающей скважины – Рдобыв = 100е5 Па, 
длина участка – Lx  = 2000 м, 
ширина участка – Ly  = 2000 м, 
толщина пласта – h = 10 м, 
диаметр скважин – d = 0.15 м, 
плотность жидкости – po = 900 кг/м3, 
упругоемкость пласта – betta = 1e–9 1/Па–1, 
абсолютная проницаемость – К = 1е–12 м2, 

динамическая вязкость жидкости – µµµµ = 1е–3 Па*с, 

так как 
µ

=σ Kh
, то гидропроводность a = 1e–8, 

максимальное число шагов по времени – Nt = 100, 
перевод добывающих скважин в нагнетательные при t > Nt/2, 
число узлов сетки по координате X – Nx = 14, 
число узлов сетки по координате Y – Ny = 14, 
координаты скважин – (1/2 Nx, 1/2 Ny), (1/4 Nx, 1/4 Ny),  
(3/4 Nx, 1/4 Ny), (3/4 Nx, 3/4 Ny), (1/4 Nx, 3/4 Ny). 

ГЛАВА 6. ЗАДАНИЯ ДЛЯ САМОСТОЯТЕЛЬНОЙ РАБОТЫ 

6.1. Вычисление дебита скважины 

Используя формулу (18) вычислить дебит центральной скважины. Вы-
вести дебит и текущее время на свободное место на листе lin . Создать еще 
одну диаграмму и организовать вывод на нее полученных значений дебита в 
зависимости от времени. 

6.2. Использование в расчетах дебита скважины 

В программе на каждом шаге задается давление на забое скважины. 
На основе формулы (13) и массива Q внести изменения в программу для 
использования дебита скважины вместо давления на забое скважины. 

6.3. Влияние неоднородности пласта на перераспределение давления 

Используя массив a, содержащий гидропроводность, показать влия-
ние неоднородности пласта на перераспределение давления по сравнению 
с однородным пластом. 

6.4. Моделирование остановки скважины 

Выводя на лист Excel значения давления на забое скважины в зависи-
мости от времени и внося соответствующие изменения в текст программы, 
смоделировать остановку скважины с получением кривой восстановления 
давления. 

6.5. Нахождение распределения гидропроводности на основе  
кривой восстановления давления 

Эта работа использует метод наименьших квадратов [9] и поиск ми-
нимума функции многих переменных на основе Поиска решения (Solver) 
в Excel. Предлагается следующая последовательность шагов: 

1) сделать необходимые изменения в программе, в частности, для ус-
корения расчетов удалить построение диаграмм; 

2) подобрать формулу, описывающую пространственное распределение 
гидропроводности и использующую независимые параметры c1, c2, …, ck. На-

пример, формула 2
321321 ),,( rcrccccc ++=σ , где 2

0
2

0 )()( yyxxr −+−= , x0 

и y0 – координаты скважины, использует параметры c1, c2, c3; 
3) для кривой восстановления давления, заданной в виде табличной 

зависимости P10(t1), P20(t2),…, Pn0(tn), рассчитать давление на скважине в 
аналогичные моменты времени P1(t1), P2(t2),…, Pn(tn) при некоторых значе-



 

35 

ниях c1, c2, …, ck. Для упрощения, вместо табличных значений можно ис-
пользовать формулу Дюпюи; 

4) вычислить сумму квадратов невязок:  

( )∑
=

−=
n

i
iiiik tPtPcccJ

1

2
021 )()(),,,( K ; 

5) используя Поиск решения (Solver) в Excel найти минимум функ-
ции многих переменных ),,,( 21 kcccJ K , задавая в качестве переменных c1, 
c2, …, ck. 

6.6. Определение границы устойчивости явного метода 

Наблюдая поведение графика найти границы устойчивости явной ко-
нечно-разностной схемы. Для этого можно использовать простейшее од-

номерное уравнение, которое получается из (12), если считать, что 0=
∂
∂

y

P  

и Q = 0: 

 2

2

x

P
a

t

P

∂
∂=

∂
∂

,  (19) 

здесь a > 0 – коэффициент, который можно положить равным единице. Явная 
конечно-разностная схема для решения уравнения (19) примет вид: 

. (20) 

И, соответственно, условие устойчивости будет: 

 
a

x
t

2

2∆δ≤∆ ,  (21) 

где коэффициент δ обеспечивает запас по устойчивости. 
Внести соответствующие изменения в программу и с точностью 10-2 

найти максимальное значение коэффициента δ в формуле (21), обеспечи-
вающее устойчивый расчет по явной схеме (20). 

6.7. Использование неявного метода и схемы прогонки 

При освоении численных методов полезно изучить метод прогонки 
[9], который используется при решении систем линейных уравнений с 
трехдиагональной матрицей коэффициентов. Применительно к уравнению 
(19) такая система линейных уравнений получается при использовании не-
явной схемы, которая имеет вид: 

 2

1
1

11
1

1 2

x

PPP
a

t

PP j
i

j
i

j
i

j
i

j
i

∆
+−=

∆
− +

−
++

+
+

.  (22) 

Записывая уравнение (22) для каждой точки i = 0,1,…,N, получим систему 
линейных уравнений: 

где α и β – стационарные граничные условия на левой и правой границах, 

∆ . Запишем систему (23) в матричной форме dPA

где a0=1, b0=0, cN=0, aN=1, ci= –r, ai=1+2r, bi= –r, i= 1,…, N–1. Видно, что 
трехдиагональная матрица имеет строгое диагональное преобладание 
|ai|>|ci|+|bi|. Это позволяет для решения системы линейных уравнений ис-
пользовать метод прогонки, который будет устойчив вне зависимости от 
величины шага интегрирования по времени [3]. 

Предлагаемый далее алгоритм метода прогонки не использует до-
полнительные переменные, но изменяет содержимое ai и di. Он имеет пря-
мой ход: 

Niadcddabcaa iiiiiiiiii K,2,1,/,/ 1111 =−=−= −−−− , 
и обратный ход: 

NN
j

N adP /1 =+
, 

0,2,1;/)( 1
1

1
K−−=−= +

+
+ NNiaPbdP i

j
iii

j
i . 

Отметим, что явная схема (20) и неявная схема (22) имеют точность 
( )2, xtO ∆∆ , поэтому при увеличении шага по времени в неявном методе точ-
ность будет снижаться по сравнению с явным методом. 

Дополнительно рекомендуется использовать аналогичный подход для ре-
шения уравнения, описывающего плоскорадиальное течение (см. раздел 1.3). 

